
Introduction 

The Holtek HT66F60A/70A are MCUs which utilise long instructions. The devices also 

have three pairs of indirect addressing pointers and registers. This allows the devices to 

access all areas of RAM without using a bank pointer BP. This application note describes 

how to access the MCU RAM using long instructions and the indirect addressing pointer 

register pairs which have the nomenclature MP1L/1H and MP2L/2H. 

Using Long Instructions to Access Data from the RAM 
Sectors 

Long instructions are used to support the full range address access for the data memory. 

When the data memory that will be accessed is located in any data memory Bank other 

than Bank 0, the long instructions can then access the data memory directly without using 

indirect addressing, which has the effect of improving CPU firmware performance. 

In this application, taking the HT66F70A as an example, it will be shown how it is possible 

to access data from different RAM Banks using long instructions and execute simple 

arithmetic operations. For the long instructions, the full RAM address must be clearly 

specified. For example, Bank 3 with an internal location of A3H will have a full address of 

03A3H. Therefore when using long instructions, the higher 8-bit address register will 

contain the bank number while the lower 8-bit address register contains the bank 

address. 

In the following program, both Bank 1 and Bank 2 RAM are accessed using long 

instructions, after which the contents of Bank 1 and Bank 2 are added together. The result 

is stored in Bank 3. Note that all long instructions begin with the letter “L”. If short 

instructions are used, there will be errors during compilation as short instructions are 

used to access data from Bank 0. 

Long instruction access data example: 

include HT66F70A.inc 

ds .section AT 080H 'data' 
cs .section 'code' 
 ORG 00H  ;HT66F70A RESET VECTOR 
MAIN: 
 MOV A,11H 
 LMOV [0180H],A  ;Move 11H to the address 80h in Bank 1 
 MOV A,22H 
 LMOV [0280H],A  ;Move 22H to the address 80h in Bank 2 
 
 LMOV A,[0180H]  ;Move the data at the address 80h in Bank 1 to ACC 
 LADD A,[0280H]  ;The data at the address 80h in Bank 1 and the data at the 
    ;address 80h in Bank 2 are added 
 LMOV [0380H],A  ;The sum result is stored at the address 80h in Bank3 
 JMP $ 

In the above example, the addresses are directly accessed. Another program which uses 

variables is now provided. Here the functions are the same as the above example. In this 

example, if short instructions are used, there are no errors during compilation, but the 

short instructions can only manipulate variables declared in Bank 0. For example, in the 

MAIN function, if the long instruction “LMOV” in line 2 “LMOV ADDRESS1, A” is changed 

to a short instruction “MOV”, the value of ADDRESS1 will remain unchanged, but the first 

address in Bank 0 will be assigned the value 11H. 

include HT66F70A.inc 

RAMBank 1 DS1   ; Declare DS1 in RAM Bank 1 
RAMBank 2 DS2   ; Declare DS2 in RAM Bank 2 
RAMBank 3 DS3   ; Declare DS3 in RAM Bank 3 
ds1 .section  AT 080H  'data' ; Declare a variables in RAM Bank 1 
ADDRESS1 DB  ? 
ds2 .section AT 080H 'data'  ; Declare a variable in RAM Bank 2 
ADDRESS2 DB  ? 
ds3 .section AT 080H 'data'  ; Declare a variable in RAM Bank 3 
ADDRESS3 DB  ? 
cs .section  'code' 
 ORG 00H   ; HT66F70A RESET VECTOR  
MAIN: 
 MOV A,11H   
 LMOV ADDRESS1,A   ; Move 11H to ADDRESS1 at address of 80h in Bank 1 
 MOV A,22H 
 LMOV ADDRESS2,A   ; Move 22H to ADDRESS2 at address of 80h in Bank 2 
 LMOV A,ADDRESS1   ; Move ADDRESS1 in Bank 1 to ACC 
 LADD A,ADDRESS2   ; Bank1 ADDRESS1 and Bank 2 ADDRESS2 are added to ACC  
 LMOV ADDRESS3,A   ; The result is saved in ADDRESS3 in Bank 3 
 JMP $  

Finally, the value of ADDRESS3 is 33H and its address is 0380H. This example has 

implemented data access from different RAM Banks. 

Use MP1L/1H, MP2L/2H to indirectly Access RAM Banks 

The HT66F70A provides five indirect memory pointers, MP0, MP1L, MP1H, MP2L and 

MP2H. These memory pointers are physically implemented in the Data Memory and are 

manipulated in the same way as normal registers providing a convenient way with which 

to address and track data. When any operation to the relevant Indirect Addressing 

Registers is carried out, the actual address that the microcontroller is directed to is the 

address specified by the related Memory Pointer. MP0, together with IAR0, are used to 

access data from Bank 0, while MP1L, MP1H and IAR1, MP2L, MP2H and IAR2 are used 

to access data from all Banks with the bank being selected by the MP1H or MP2H 

registers. The MP1H and MP2H registers point to the RAM Bank number and play the 

same role as a bank pointer. 

This following example takes the HT66F70A as an example and describes how to move 

data 11H and 22H to Bank 1 RAM and Bank 2 RAM respectively using indirect address 

instructions and then exchange the contents of the Bank 1 and Bank 2. It can be seen 

that for MCUs which do not have long instructions, data manipulation is more complex 

when it must be implemented using the indirect addressing pairs. 

In this example MP1L, MP1H and IAR1, MP2L, MP2H and IAR2 are used to access RAM: 

include HT66F70A.inc 

RAMBank 1 DS1   ;Declare DS1 in RAM Bank 1 
RAMBank 2 DS2   ;Declare DS2 in RAM Bank 2 
ds .section AT 080H 'data'  ; Variable default in RAM Bank 0 
BLOCK  DB  ? 
NUM  DB  ? 
ds1  .section AT 080H  'data' ; Declare variable in RAM Bank 1 
ADDRESS1 DB ? 
ds2  .section AT 080H  'data' ;Declare variable in RAM Bank 2 
ADDRESS2 DB  ? 
cs .section  'code' 
 ORG 00H   ;HT66F70A RESET VECTOR 
MAIN: 
 MOV  A,080H       
 MOV  BLOCK,A    ;In one Bank RAM, there are totally 080H=128 bytes 
 MOV  A,01H 
 MOV  MP1H,A        ;MP1H=01H, set MP1L/MP1H,IAR1 to access data from Bank1 
     ;indirectly 
 MOV  A,02H 
 MOV  MP2H,A   ;MP2H=01H, set MP2L/MP2H,IAR2 to access data from Bank2 
     ;indirectly 
 MOV  A,OFFSET ADDRESS1 
 MOV  MP1L,A   ;The address of variable ADDRESS1 in Bank 1, that’s Bank1 
     ;start address [0180h] 
 MOV  A,OFFSET ADDRESS2 
 MOV  MP2L,A   ;The address of variable ADDRESS2 in Bank 2, that’s Bank2 
     ;start address [0280h] 
LOOP:    ;The function of LOOP is to move 011h to entire RAM Bank 1 
     ;and 022h to entire RAM Bank 2 
 MOV  A,011H 
 MOV  IAR1,A   ;Move 011h to Bank 1 
 MOV  A,022H 
 MOV  IAR2,A   ;Move 022h to Bank 2 
 INC   MP1L    ;Bank 1 pointer plus 1 
 INC   MP2L   ;Bank 2 pointer plus 1 
 SDZ  BLOCK 



Using the HT66F60A/HT66F70A with Long Instructions 
to Access Data from RAM Other than Bank 0 

4 

 JMP LOOP 
 MOV A,080H 
 MOV BLOCK,A  
 DEC MP1L  
 DEC  MP2L      
LOOP1:    ;The function of LOOP1 is to exchange the data of entire 
     ;RAM Bank1 and Bank2 
      
 LMOV  A,IAR1    
 LMOV NUM,A   ;Move the data from Bank1 to the middle variable NUM 
 LMOV A,IAR2    
 LMOV  IAR1,A   ;Move the data from Bank2 to Bank1 
 LMOV  A,NUM 
 LMOV IAR2,A   ;Move the middle variable data from Bank1 to Bank2 
 DEC MP1L    ;Bank1 pointer sub 1 
 DEC  MP2L   ;Bank2 pointer sub 1 
 SDZ  BLOCK 
 JMP LOOP1 
 JMP $   ;End 

After executing LOOP, the value will be 11H in Bank 1 and 22H in Bank 2, while after executing 

LOOP1, their values are interchanged with 22H in Bank 1 and 11H in Bank 2. 

Conclusions 

This application has described how to carry out direct access on every RAM Bank using long 

instructions as well as describing how to carry out indirect access on every RAM Bank using the 

MP1L/1H and MP2L/2H memory pointer register pairs. The different examples have shown how 

RAM access to Banks other than Bank 0 is a relatively easy process.  

Appendix 

Refer to the attached file on the website for Source Code details. 


